County Science Fair 2006

How temperature affects capillary action

Abe Karplus
4th grade • Spring Hill School • March 11, 2006

What my experiment is about

My experiment is about how the temperature of water in a jar will affect how high the water will rise in a paper towel due to the phenomenon known as capillary action. My second experiment is about how the temperature of water in a jar will affect how long it takes the water to rise a certain distance in a paper towel.

Theory

For my experiment's theory, I used Washburn's equation [1], which I got from Wikipedia. Washburn's equation is

$$L^2 = (y\ D\ t) \div (4\ n)$$

in which L is length in meters, y is surface tension of the liquid in N m^{-1}, D is average pore diameter of the material in meters, t is time in seconds, and n is viscosity of the liquid in N s m^{-2}. The variables that change with temperature are y and n [2,3], but we can look them up on the web. We choose the value of t, and we measure the value of L. Using these allows us to compute the value of D.

Using the length from one of my preliminary experiments (5.5 cm at 20°C), I estimated D to be 0.0000014 m (1.4 microns) using

$$D = (4\ n\ L^2) \div (y\ t)\ .$$

variable	value
t	120 s
y	0.0728 N m^{-1}
n	0.001002 N s m^{-2}
L	0.055 m

variable	value
D	0.0000014 m

At 20°C, I measured L to be 0.055 m. But what would L be at different temperatures? I used Washburn's equation:

$$L=\sqrt{((y\ D\ t) \div (4\ n))}$$

to find out:

temperature in °C	y in $N\ m^{-1}$	n in $N\ s\ m^{-2}$	L in m
0	0.076	0.001792	0.042
40	0.069	0.000653	0.066
60	0.065	0.000467	0.076

Hypothesis

My hypothesis is that the hotter the water is, the higher it will rise on the paper towel.

Materials

For this experiment I used

- a jar half-filled with water
- strips of paper towel (1.5 cm by 28.5 cm) with a line marked on them in pen 5 cm from one end
- a ruler
- a thermometer and clip for holding it onto the jar
- a microwave oven (for heating water)
- a stopwatch
- a paperclip or other weight
- ice cubes.

Procedure for data collection

1. Prepare the strip by
   a) drawing a line 5 cm from the end and
   b) putting a weight on that end of the strip.
2. Prepare the water by
   a) putting it in a microwave oven or
   b) putting it in the refrigerator.
3. Measure and record water temperature.
4. Dip strip into water to line mentioned in 1.a) for 2 minutes, then remove.
5. Measure and record
   a) water distance above the line and
   b) water temperature
6. Repeat as many times as necessary.

Results

temp °F	L (cm)
47.5	5
48	4.7
50	5
60	5.0
60	4.75
60	4.5
68	5.5
68	5.5
104	5.5
123	5.5
127	5.8
131	6.0
148	6.1
160	6.6
161	5.5
163	6.75

Procedure for data analysis

1. Make a table (not plot) of length and temperature, using the average of the before and after temperatures.

2. Use the plotting program gnuplot to make a plot of length vs. temperature.

3. Also using gnuplot, fit a straight line to the dots. Fitting a straight line means putting in a straight line where there is the shortest difference in length (y) between the line and the points. Here is the script I used to do this:

```
set terminal postscript eps color solid lw 2 24
set output 'temp-length.eps'
set xlabel 'temperature (F)'
set ylabel 'distance (cm)'
unset key
set title 'How temperature affects capillary action'
fit m*x+b 'temp-length-table.txt' via m,b
plot 'temp-length-table.txt', m*x+b
```



The plot showed me that my hypothesis was correct—on average the greater the temperature gets, the greater the length gets. But is the theory correct?

4. Add two new columns to the table in 1: temperature in °C and viscosity [3].

5. Use gnuplot to find the value of D that makes Washburn's equation fit the data best. Using my data, the best fit for D is $9.8 * 10^{-7}$ m, or nearly one micron.

6. Using this value of D, plot a graph for what Washburn's equation says should happen (using the following gnuplot script).

```
set terminal postscript eps color solid lw 2 20
set output 'washburn.eps'
set xlabel 'temperature (F)'
set ylabel 'distance (cm)'
unset key
set title "How Washburn's equation says temperature affects capillary action"

# Washburn's equation, with x=surface tension/viscosity
# L in cm
L(D,t,x) = 100 * sqrt(D*t*x/4)

# line for surface tension,
# from hyperphysics.phy_astr.gsu.edu/hspace/surten.html
# in N m^(-1)
y(tc) = (76.5 -0.185*tc) * 0.001

D=0.000001
fit L(D,120,x) 'temp-length-table.txt' using (y($3)/$4):($2) via D
plot 'temp-length-table.txt', \
       'temp-length-table.txt' using 1:(L(D,120,(y($3)/$4))) with lines
print D
```



This plot’s line is slightly steeper than the previous one’s. This means that as temperature increases the height of the water should increase more rapidly than it does. However, seeing as Washburn’s equation does not take gravity into account, it is not as accurate as a formula that would. Also, since a taller column of water weighs more than a shorter column of water, the effect of gravity is greater the taller the column, making the effect of temperature smaller.

Second experiment

I measured how long it takes for water to rise a fixed distance at different temperatures. Thus the effect of gravity was equal for all of these and should cancel out.

The differences in procedure from the first experiment are the following:

* In step 1c, draw another line 3.5 cm above the the line at 5 cm.
* In step 4, dip strip in water to the lower line and time how long it takes for the water to reach the upper line.
* In step 5a, record time in seconds.

Each of the 31 rows in this table of my results is one experiment. The first column is the average of the starting and ending temperatures in degrees Fahrenheit. The second column is time in seconds. The third column is temperature in degrees Celsius converted from the first column with the formula C=(F-32)*5÷9. The fourth column is the viscosity of water at the specified temperature derived from the table at www.thermexcel.com/english/tables/eau_atm.htm For the temperatures between those in the table, I had to interpolate a bit.

temp °F	time (sec)	temp °C	viscosity (N s m^{-2})
168.5	18	75.8	0.000373
164.5	22	73.6	0.000385
162.5	17	72.5	0.000391
160	14	71.1	0.000399
157	15	69.4	0.000406
153.5	17	67.5	0.000419
150	14	65.6	0.000431
147	20	63.9	0.000440
143	22	61.7	0.000455
140	20	60	0.000467
136	31	57.8	0.000481

temp °F	time (sec)	temp °C	viscosity (N s m^{-2})
132.5	23	55.8	0.000496
126	19	52.2	0.000529
113	25	45	0.000596
110	36	43.3	0.000615
110	33	43.3	0.000615
99.5	43	37.5	0.000685
98.5	34	36.9	0.000692
97	38	36.1	0.000705
85	44	29.4	0.000807
84	63	28.9	0.000815
83	43	28.3	0.000835
74	49	23.3	0.000930
73.5	54	23.1	0.000933
73.5	68	23.1	0.000933
61	71	16.1	0.001109
59.5	64	15.3	0.001135
59.5	76	15.3	0.001135
42	144	5.6	0.001497
40.3	118	4.6	0.001545
36.5	183	2.5	0.001545

Data analysis for second experiment

I rewrote the gnuplot script from the first experiment into

```
set terminal postscript eps color solid lw 2 24
set output 'temp-time.eps'
set xlabel 'temperature (F)'
set ylabel 'time (s)'
unset key
set title 'How temperature affects capillary action'
fit m*x+b 'temp-time-table.txt' via m,b
plot 'temp-time-table.txt', m*x+b
```

to get the following plot:



One strange thing about this plot is that it goes into negative time. The green line is the best straight-line fit for my data, but since it goes into negative time at high temperatures, it is clearly impossible. How could the water rise up to the second line before it even touched the first line?

Washburn's equation can be changed into

$$\mathbf{t = 4\ n\ (L/100)^2 / (y\ D)}$$

The reason for the L/100 is to convert centimeters into meters. However, gnuplot is designed to fit only functions of one variable. Both y and n vary with temperature, so we have to introduce a new variable

$$\mathbf{x = y/n\ .}$$

Because our equation has n/y instead of y/n, we have to put x on the bottom, not the top:

$$\mathbf{t = 4\ (L/100)^2 / (x\ D)\ .}$$

I changed the previous Washburn script into

```
set terminal postscript eps color solid lw 2 20
set output 'washburn-time.eps'
set xlabel 'temperature (F)'
set ylabel 'time (s)'
unset key
set title "How Washburn's equation says temperature affects capillary action"

# Washburn's equation, with x=surface tension/viscosity
# L in cm
L(D,t,x) = 100 * sqrt(D*t*x/4)
t(D,x,L) = 4*(L/100)**2/(x*D)

# line for surface tension,
# from hyperphysics.phy_astr.gsu.edu/hspace/surten.html
# in N m^(-1)
y(tc) = (76.5 -0.185*tc) * 0.001

D=0.001
fit t(D,x,3.5) 'temp-time-table.txt' using (y($3)/$4):($2) via D

print D
plot 'temp-time-table.txt', 'temp-time-table.txt' \
      using 1:(t(D,y($3)/$4,3.5)) with lines
```

to get D= $9.8 * 10^{-7}$ m and the following plot:



The two strange things are that Washburn's equation says that there should less effect than we measured (not more, like in the first experiment) and D comes out the same in both experiments (this good, but surprising).

Conclusion

My hypothesis was correct—on average the hotter the water gets, the higher it rises on the paper towel. However, Washburn's equation is not a very good fit for our data.

Future work

If I ever repeat this experiment, a few things I might change are

- Design a Lego machine to automate the measuring of how long the water takes to go a certain distance.
- Do the experiment again with different materials to determine their average pore diameter (D).
- Experiment with different liquids instead of water. This would make the theory harder, since a web site is less likely to have surface tension and viscosity tables for, say, milk instead of water.

Acknowledgments

My dad, Kevin Karplus, gave me a few suggestions on how to do my experiment, helped me find stuff on the web, explained Washburn's equation and how to use gnuplot, and typed up my report and poster (taking dictation).

References

1. Wikipedia Washburn's equation http://en.wikipedia.org/wiki/Washburn%27s_equation

2. Surface Tension http://hyperphysics.phy-astr.gsu.edu/Hbase/surten.html

3. Physical characteristics of water (at the atmospheric pressure) http://www.thermexcel.com/english/tables/eau_atm.htm

Parental Comments

This project was an ambitious one for a 4th grader, so some parental help was needed, though perhaps not as much as a casual observer might assume. The initial idea of looking at capillary action and how temperature affects it was all his —I was worried that the effect might be too small to be measured with the crude equipment we had.

The experiment design was mostly his. I suggested some of the preliminary experiments that were used to decide what absorbent material to use (muslin, coffee filter, paper towel). I also suggested that he vary only the water temperature, as the temperature of the material would be too hard to measure. I suggested the second experiment (measuring time instead of length) when the first one showed a smaller result than expected.

I also suggested the use of a clamp when he complained about his arm getting tired, and drilled holes in a clothespin so that it could be attached to Lego Technic pieces. He built a clamp using the clothespin, but ended up finding it too hard to use (friction between the paper and the pin was too high), so he went back to holding the strips by hand.

I helped him find material on the web about capillary action and the effects of temperature on viscosity and surface tension of water, as much of the material was written for physicists or chemical engineers.

He did the preliminary theoretical calculations on his pocket calculator, and I did the same set of calculations on my calculator. We found one discrepancy in the computations, but the error was in my data entry, not in his computations.

He needed three hands for his experiment, so I (or his classmate) started the stopwatch for him, and sometimes counted down the last few seconds of the 2-minute period. I also took photos of him for the poster.

I wrote the gnuplot scripts to do the plotting and the data fitting for the first experiment, but explained them carefully to him. He did the gnuplot scripts for the second experiment, though it took him a few tries to get the algebra right.

I typed the report and poster, taking dictation. The layout of the poster is Abe's, though he did ask for suggestions on where to put the photos.

—Kevin Karplus